

**Prototype Alat Penyortiran Buah Tomat dengan Tingkat
Kematangan berdasarkan warna menggunakan Sensor Warna
TCS3200**



Disusun Oleh :

Abdul Gafur Rehalat (210102D001)

Habibi La Marno (210102D119)

Reyhandayuni Arum Saputra (210102D129)

Zubaidah Usemahu (210102D137)

**PROGRAM STUDI SISTEM INFORMASI
INSTITUT TEKNOLOGI DAN BISNIS STIKOM AMBON**

Kata Pengantar

Puji syukur kehadirat Allah SWT atas segala limpahan rahmat, taufik, dan hidayah-Nya sehingga penulis dapat menyelesaikan laporan dengan judul “Prototype Alat Penyortiran Buah Tomat Berdasarkan Tingkat Kematangan Menggunakan Sensor Warna TCS3200” dengan baik dan tepat waktu.

Laporan ini disusun sebagai salah satu bentuk dokumentasi dari kegiatan perancangan dan implementasi prototype sistem penyortiran buah tomat berbasis sensor warna. Tujuan dari pembuatan prototype ini adalah untuk memberikan gambaran mengenai pemanfaatan teknologi sensor dan mikrokontroler dalam bidang pertanian, khususnya dalam membantu proses pemilahan buah tomat berdasarkan tingkat kematangannya.

Dalam penyusunan laporan ini, penulis menyadari masih banyak kekurangan, baik dari segi materi maupun teknis penulisan. Oleh karena itu, kritik dan saran yang membangun dari berbagai pihak sangat penulis harapkan demi kesempurnaan laporan ini.

Ucapan terima kasih penulis sampaikan kepada semua pihak yang telah memberikan dukungan, bimbingan, dan bantuan, baik secara langsung maupun tidak langsung, sehingga laporan ini dapat terselesaikan. Secara khusus penulis mengucapkan terima kasih kepada dosen pembimbing, rekan-rekan, serta pihak lain yang turut membantu dalam proses penelitian dan penyusunan laporan.

Akhir kata, semoga laporan ini dapat memberikan manfaat bagi pembaca, serta menjadi referensi bagi penelitian atau pengembangan selanjutnya dalam bidang sistem otomatisasi berbasis sensor.

Ambon, 18 September 2025

Penyusun

Abstrak

Perkembangan teknologi saat ini membuka peluang untuk mengatasi berbagai permasalahan dalam industry pertanian, salah satunya adalah proses penyortiran buah tomat berdasarkan tingkat kematangan. Selama ini, penyortiran tomat masih banyak dilakukan secara manual, yang rentan terhadap kesalahan akibat subjektivitas manusia. Penelitian ini bertujuan untuk merancang dan membangun prototipe alat Arduino Uno R3. Sensor TCS3200 dipilih karena bersifat open source, mudah diprogram, dan mampu mendeteksi perbedaan warna tomat yang mencerminkan Tingkat kematangannya yaitu mentah dan matang. System ini diharapkan dapat melakukan penyortiran secara otomatis dan meningkatkan efisiensi serta akurasi dalam proses sortir. Metode yang digunakan mencakup perancangan prototipe, pemrograman deteksi warna dan pengujian akurasi system. Hasil dari penelitian ini diharapkan dapat memberikan kontribusi nyata dalam modernisasi sektor pertanian melalui pemanfaatan teknologi tepat guna.

Kata kunci : Penyortiran tomat, sensor warna TCS3200, Arduino Uno R3, Tingkat kematangan, otomatis pertanian

Abstract

The advancement of technology today offers solutions to various challenges in the agricultural sector, including the sorting of tomatoes based on their ripeness. Traditionally, tomato sorting is done manually, which is time-consuming and prone to errors due to human subjectivity. This study aims to design and develop a prototype for automatic tomato sorting using the TCS3200 color sensor and Arduino Uno R3 microcontroller. The TCS3200 sensor was chosen for its open-source nature, ease of programming, and ability to distinguish color variations that represent different ripeness levels: unripe, semi-ripe, and fully ripe. The system is expected to perform automatic sorting to improve efficiency and accuracy in the sorting process. The research method includes designing the prototype, programming the color detection system, and testing its accuracy. The results of this study are expected to contribute to the modernization of agriculture through appropriate technology.

Keywords: Tomato sorting, TCS3200 color sensor, Arduino Uno R3, ripeness level, agricultural automation

DAFTAR ISI

JUDUL

KATA PENGANTAR.....	i
---------------------	---

ABSTRAK.....	ii
--------------	----

BAB I PENDAHULUAN

1.1 Latar Belakang.....	1
1.2 Rumusan Masalah.....	2
1.3 Tujuan	2
1.4 Alasan Penelitian.....	3

BAB II MATERI DAN METODE

2.1 Objek Penelitian.....	4
2.2 Lokasi dan Waktu Penelitian.....	4
2.3 Alat dan Bahan.....	4-7
2.4 Metodologi Penelitian	7-20
2.5 Teknik Analisis Data.....	21

BAB III HASIL DAN PEMBAHASAN

3.1 Hasil Pengujian Alat.....	22
3.2 Pembahasan.....	22
3.3 Pengujian Blackbox.....	24-26

BAB IV PENUTUP

4.1 Kesimpulan.....	27
4.2 Saran.....	27

DAFTAR PUSTAKA

BAB I

PENDAHULUAN

1.1 Latar Belakang

Perkembangan ilmu pengetahuan dan teknologi dewasa ini telah memberikan banyak kemudahan dalam berbagai aspek kehidupan, termasuk dalam bidang pertanian. Salah satu tantangan yang dihadapi dalam sektor pertanian adalah proses penyortiran hasil panen yang masih banyak dilakukan secara manual. Penyortiran secara manual memiliki kelemahan, antara lain ketergantungan pada ketelitian manusia, waktu proses yang relatif lama, serta potensi kesalahan dalam klasifikasi yang tinggi akibat faktor subjektivitas.

Buah tomat merupakan salah satu komoditas hortikultura yang banyak dibudidayakan dan dikonsumsi di Indonesia. Tingkat kematangan tomat sangat memengaruhi kualitas dan nilai jualnya. Oleh karena itu, proses penyortiran tomat berdasarkan tingkat kematangan menjadi tahap penting sebelum produk didistribusikan ke pasar. Warna kulit tomat dapat dijadikan sebagai indikator visual yang membedakan antara tomat matang, setengah matang, dan mentah.

Dengan kemajuan teknologi, kini telah tersedia berbagai sensor dan mikrokontroler yang dapat dimanfaatkan untuk merancang sistem penyortiran otomatis. Salah satunya adalah sensor warna TCS3200 yang mampu membaca nilai RGB dari objek yang dikenainya. Sensor ini dapat dihubungkan dengan mikrokontroler seperti Arduino Uno R3 untuk mengolah data warna dan mengklasifikasikan tomat berdasarkan tingkat kematangan.

Selain itu, sistem ini juga dilengkapi dengan sensor infrared yang berfungsi untuk mendeteksi keberadaan objek (tomat) pada jalur sortir, relay sebagai saklar elektronik untuk mengontrol motor DC, serta motor servo untuk mengarahkan buah ke tempat sortir yang sesuai. Informasi klasifikasi tomat kemudian ditampilkan melalui LCD sebagai tampilan keluaran.

Berdasarkan uraian tersebut, maka penulis tertarik untuk melakukan penelitian dengan judul:

"Prototipe Penyortiran Buah Tomat dengan Tingkat Kematangan berdasarkan warna dengan sensor TCS3200", yang bertujuan untuk merancang sebuah sistem sortir tomat otomatis berbasis mikrokontroler sebagai solusi tepat guna yang dapat meningkatkan efisiensi, akurasi, dan konsistensi dalam proses penyortiran.

1.2 Rumusan Masalah

Berdasarkan identifikasi masalah diatas dapat di tentukan rumusan masalah sebagai berikut:

1. Bagaimana merancang protitpe buah tomat dengan menggunakan sensor warna TCS3200?
2. Bagaimana melakukan penyortiran otomatis yang akan membawa buah tomat ke sensor warna tersebut?
3. Bagaimana cara sensor warna mendeteksi Tingkat kematangan buah?

1.3 Tujuan Penelitian

1. Mengembangkan system deteksi warna yang mampu membedakan tomat matang dan mentah secara akurat.
2. Membuat rancangan prototipe yang dapat mendeteksi dan memisahkan buah tomat yang matang atau mentah menggunakan warna sebagai parameter.
3. Melakukan Analisa terhadap perubahan warna dari alat sensor dan akurasi system untuk deteksi warna tomat sebagai objek.
4. Membuat prototipe pesortir buah tomat berbasis Arduino Uno R3 sebagai mikrokontroler

1.4 Tinjauan Pustaka

Penelitian mengenai sistem penyortiran buah berbasis teknologi telah banyak dikembangkan untuk meningkatkan efisiensi dan akurasi dalam proses sortasi hasil pertanian. Salah satu penelitian terkait dilakukan oleh Amin, Akbar, dan Widasari [1] yang merancang sistem sortir buah apel menggunakan kombinasi sensor warna dan sensor suhu. Dalam penelitian tersebut, sensor warna digunakan untuk mengklasifikasikan warna kulit buah apel sebagai indikator kematangan, sementara sensor suhu membantu dalam mendeteksi kondisi kesegaran buah. Mikrokontroler digunakan sebagai pusat kendali untuk mengolah data dari sensor, kemudian mengaktifkan aktuator seperti motor servo untuk memindahkan buah ke tempat sortir yang sesuai.

Hasil dari penelitian ini menunjukkan bahwa sistem mampu bekerja secara otomatis dan cukup akurat dalam memilah buah berdasarkan kriteria yang telah ditentukan. Penelitian ini menjadi dasar penting dalam pengembangan sistem penyortiran otomatis, termasuk untuk buah lain seperti tomat, dengan pendekatan serupa menggunakan sensor warna dan kontrol berbasis mikrokontroler.

1.5 Alasan Penelitian

Penelitian ini penting dilakukan karena proses penyortiran buah tomat berdasarkan tingkat kematangan merupakan tahapan krusial dalam menjaga kualitas produk sebelum didistribusikan ke pasar. Selama ini, proses sortir masih banyak dilakukan secara manual, yang tidak hanya memerlukan waktu dan tenaga yang besar, tetapi juga memiliki tingkat akurasi yang rendah akibat subjektivitas manusia. Hal ini dapat menyebabkan tercampurnya tomat dengan tingkat kematangan yang berbeda, sehingga berdampak pada kepuasan konsumen dan penurunan nilai jual.

Dengan memanfaatkan teknologi sensor warna TCS3200 dan mikrokontroler, proses penyortiran dapat dilakukan secara otomatis, cepat, dan konsisten. Penggunaan alat otomatis juga dapat membantu para petani, distributor, dan pelaku usaha pertanian dalam meningkatkan efisiensi kerja dan kualitas produk. Selain itu, penelitian ini juga berkontribusi dalam pengembangan teknologi tepat guna di bidang pertanian, yang sangat dibutuhkan dalam menghadapi tantangan modernisasi sektor pertanian di Indonesia.

BAB II

MATERI DAN METODE

2.1 Objek Penelitian

Objek penelitian ini adalah buah tomat yang diklasifikasikan berdasarkan Tingkat kematangannya menggunakan alat prototipe otomatis yang dirancang dengan sensor warna TCS3200. Alat ini berfungsi untuk mendeteksi warna permukaan tomat, kemudian mengelompokkan tomat ke dalam kategori mentah (hijau) dan matang (merah).

2.2 Lokasi Dan Waktu Penelitian

Penelitian ini dilakukan secara mandiri oleh peneliti di rumah. Seluruh proses perancangan dan pembuatan alat, termasuk perakitan komponen elektronik, pemrograman mikrokontroler, serta pengujian alat, dilakukan secara manual dan menggunakan alat bantu sederhana yang tersedia di rumah.

Waktu pelaksanaan penelitian berlangsung selama bulan Mei hingga Agustus 2025. Pengujian dilakukan di dalam ruangan dengan pencahayaan yang dikondisikan secara sederhana untuk mendukung pembacaan sensor warna secara optimal. Tomat yang digunakan memiliki ukuran dan jenis yang relatif seragam agar hasil pengujian lebih akurat dan konsisten.

2.3 Alat dan Bahan


2.3.1 Spesifikasi Perangkat Keras (Hardware)

- Laptop Dell G5 15

Laptop digunakan sebagai pusat pengolahan data dan media untuk melakukan pemrograman. Laptop memiliki kemampuan yang cukup untuk menjalankan software Arduino IDE, aplikasi pemrograman, serta mendukung proses pengujian dan monitoring system.

- Sensor Warna TCS3200

Sensor ini dapat membaca nilai RGB (Red, Blue, Green) dari Cahaya yang dipantulkan objek, kemudian dikirimkan ke mikrokontroler untuk dianalisis dan digunakan dalam proses pengambilan Keputusan.



gambar 1 : sensow warna TCS3200

- Arduino Uno R3

Arduino Uno R3 merupakan mikrokontroler utama yang mengontrol seluruh system. Perangkat ini bertugas menerima input dari sensor, kemudian memproses data tersebut, dan memberikan output berupa kendali ke aktuator (motor DC, motor servo, relay, maupun tampilan ke LCD).



gambar 2 : Arduino Uno R3

- Motor DC

Motor DC digunakan sebagai penggerak mekanik pada system, misalnya untuk memutar konveyor.



gambar 3 : motor DC

- Motor Servo

Motor servo berfungsi sebagai actuator yang membutuhkan pergerakan presisi. Biasanya digunakan untuk menggerakkan lengan atau tuas sortir yang memindahkan objek berdasarkan hasil pembacaan sensor.



gambar 4 : motor servo

- LCD

LCD digunakan untuk menampilkan informasi kepada pengguna, misalnya status system, hasil pembacaan sensor, maupun data proses penyortiran. LCD memudahkan pengguna untuk melakukan monitoring secara langsung.



gambar 5 : LCD

- Sensor Infrared


Sensor infrared digunakan untuk mendeteksi adanya objek pada jalur system. Fungsinya biasanya untuk mengetahui posisi atau keberadaan benda sebelum dilakukan pembacaan sensor warna atau penyortiran.



gambar 6 : sensor IR

- Relay

Relay berfungsi sebagai satu elektronik yang dikendalikan oleh Arduino. Relay biasanya digunakan untuk mengatur aliran arus Listrik ke komponen tertentu, seperti motor DC, sehingga Arduino tetap aman dari beban arus besar.



gambar 7 : relay

2.3.2 Spesifikasi Perangkat Lunak (Software)


- System Operasi : Windows
- Bahasa Pemrograman : C++
- Tools yang digunakan : Arduino IDE, draw.io, Fritzing

2.4 Metodologi Penelitian

Penelitian ini menggunakan metode Research and Development (R&D), yaitu dengan merancang, membuat dan menguji alat penyortiran tomat otomatis menggunakan sensor warna dan mikrokontroler Arduino Uno R3. Tahapan metodologi penelitian dilakukan sebagai berikut:

2.4.1 Analisis sistem yang sedang berjalan


Sebelum alat penyortir buah tomat ini dibuat, proses penyortiran masih dilakukan secara manual oleh manusia atau petani. Cara ini membutuhkan waktu yang cukup lama dan memerlukan tenaga ekstra. Sistem penyortiran yang sedang berjalan saat ini belum menggunakan alat bantu otomatis. Gambaran sistem kerja manual tersebut dapat dilihat pada Gambar



2.4.2 Perancangan system yang diusulkan


Konsep dari prototipe alat penyortir buah tomat ini bertujuan untuk menggunakan teknologi untuk membantu dan mempermudah proses penyortiran buah tomat. Ketika buah tomat diletakkan pada alat ini, system akan secara otomatis mendeteksi warna tomat dan menyortirnya sesuai dengan Tingkat kematangannya. Untuk menjelaskan kerja system secara lebih jelas, maka dibuat beberapa model perancangan berikut :

- Data Flow Diagram (DFD)



gambar 8 : DFD

- Flowchart



gambar 9 : Flowchart

Proses penyortiran buah otomatis dimulai ketika sistem diaktifkan dan conveyor mulai berjalan. Buah yang masuk ke jalur akan terlebih dahulu diperiksa oleh sensor infrared pertama (IR1). Jika IR1 tidak mendeteksi adanya buah, conveyor tetap berjalan dan sistem kembali menunggu buah berikutnya. Namun, jika IR1 mendeteksi buah, maka conveyor akan berhenti sejenak agar proses pendeteksian dapat dilakukan dengan akurat.


Saat conveyor berhenti, sensor warna akan membaca warna buah. Informasi hasil pembacaan ini kemudian ditampilkan pada layar LCD, sehingga pengguna dapat mengetahui warna yang terdeteksi. Setelah itu, conveyor kembali bergerak membawa buah menuju tahap sortir.

Selanjutnya, sistem membuat keputusan berdasarkan warna buah. Jika sensor warna mendeteksi buah merah, maka buah diteruskan ke sensor IR2 untuk divalidasi. Setelah terdeteksi oleh IR2, conveyor tetap berjalan tanpa mengaktifkan servo, sehingga buah merah lewat langsung di jalur utama conveyor.

Sebaliknya, jika sensor warna mendeteksi buah hijau, maka buah juga divalidasi oleh IR2. Setelah posisi buah hijau terdeteksi, sistem akan memberikan perintah pada servo untuk aktif. Servo kemudian mendorong buah hijau keluar dari conveyor ke tempat penampungan khusus.

Setelah proses penyortiran selesai, sistem kembali ke kondisi awal dan siap memproses buah berikutnya. Dengan demikian, sistem ini bekerja secara berulang untuk memisahkan buah berdasarkan warnanya, di mana buah merah tetap berjalan di conveyor utama, sementara buah hijau dipisahkan dengan bantuan servo.

- Rangkaian Wiring



gambar 10 : fritzing

2.4.3 Perakitan dan pembuatan alat

Semua komponen dirakit secara manual, proses ini mencakup :

- Penyambungan kabel dan koneksi antar komponen ke Arduino Uno R3, Masing-masing komponen dihubungkan ke pin digital dan analog sesuai fungsi, seperti input sensor dan output aktuator. Wiring dilakukan menggunakan jumper kabel dan breadboard untuk kemudahan pengujian awal.
- Pemasangan sensor IR sebagai pendeteksi keberadaan tomat. Sensor ini dipasang pada jalur masuk buah tomat untuk mendeteksi apakah ada objek yang akan diproses. Sinyal dari sensor digunakan sebagai trigger untuk memulai pembacaan warna.
- Pemasangan sensor warna sebagai pendeteksi warna tomat, sebagai pendeteksi tingkat kematangan tomat. Sensor ini akan membaca nilai RGB permukaan buah, kemudian sistem akan memproses hasil pembacaan untuk menentukan kategori (matang atau mentah).
- Penggunaan relay untuk mengontrol motor DC, sebagai penggerak konveyor. Relay berfungsi sebagai saklar elektronik yang menghubungkan motor DC dengan catu daya. Arduino akan mengaktifkan relay sesuai kebutuhan agar konveyor bergerak.
- Instalasi motor servo untuk mekanisme sortir, sebagai penggerak mekanisme sortir. Motor servo digunakan untuk menggerakkan tuas sortir agar buah tomat dapat diarahkan ke tempat yang sesuai berdasarkan kategorinya.
- Pemasangan LCD 12C 16x2, untuk menampilkan hasil deteksi warna dan status sortir. LCD dipasang pada posisi yang mudah dilihat oleh operator dan menampilkan informasi seperti warna terdeteksi dan status penyortiran buah.

2.4.4 Pemograman Mikrokontroler

```
#include <Wire.h>
#include <LiquidCrystal_I2C.h>
#include <Servo.h>
#include <math.h>

// LCD & Servo
LiquidCrystal_I2C lcd(0x27, 16, 2);
Servo servoSortir;

// Pin
#define SERVO_SORTIR 5
#define S0 8
#define S1 9
#define S2 10
#define S3 11
#define salidaTCS 12
#define IR_SENSOR1 6
#define IR_SENSOR2 4
#define RELAY_CONVEYOR 7

// Posisi servo
const int POS_SORTIR_AWAL = 0;
const int POS_SORTIR_BUANG = 60;

// Counter
int countMasak = 0;
int countMentah = 0;

// Servo non-blocking
bool servoBusy = false;
unsigned long servoStartTime = 0;
const unsigned long servoDurasi = 500;
```



```

// Status buah saat ini
bool buahSaatIniMerah = false;
bool menungguBuahDiIR2 = false;

// Variabel untuk mendeteksi transisi IR2
bool prevIR2State = false;

// Waktu conveyor berhenti
const unsigned long conveyorStopTime = 750;

// ===== Threshold Berdasarkan Kalibrasi =====

struct ColorThresholds {
    int avgRedR = 521, avgRedG = 1228, avgRedB = 1199;
    int avgGreenR = 485, avgGreenG = 570, avgGreenB = 564;
};

ColorThresholds thresholds;

// ===== Setup =====

void setup() {
    pinMode(S0, OUTPUT);
    pinMode(S1, OUTPUT);
    pinMode(S2, OUTPUT);
    pinMode(S3, OUTPUT);
    pinMode(salidaTCS, INPUT);
    pinMode(IR_SENSOR1, INPUT);
    pinMode(IR_SENSOR2, INPUT);
    pinMode(RELAY_CONVEYOR, OUTPUT);

    digitalWrite(S0, HIGH);
    digitalWrite(S1, LOW);
    digitalWrite(RELAY_CONVEYOR, HIGH);

```

```

Serial.begin(9600);

servoSortir.attach(SERVO_SORTIR);
servoSortir.write(POS_SORTIR_AWAL);


lcd.init();
lcd.backlight();
lcd.clear();
lcd.setCursor(0, 0);
lcd.print("Merah : 0");
lcd.setCursor(0, 1);
lcd.print("Hijau : 0");


Serial.println("=== SISTEM SORTING BUAH SIAP ===");
}

// ===== Fungsi Warna =====

void bacaWarnaRaw(int &r, int &g, int &b) {
    digitalWrite(S2, LOW);
    digitalWrite(S3, LOW);
    r = pulseIn(salidaTCS, LOW, 50000);
    digitalWrite(S2, HIGH);
    digitalWrite(S3, HIGH);
    g = pulseIn(salidaTCS, LOW, 50000);
    digitalWrite(S2, LOW);
    digitalWrite(S3, HIGH);
    b = pulseIn(salidaTCS, LOW, 50000);
}

void bacaWarna(int &r, int &g, int &b) {
    const int numSamples = 10;
    long sumR = 0, sumG = 0, sumB = 0;
    int validSamples = 0;

    for (int i = 0; i < numSamples; i++) {

```

```

int tempR, tempG, tempB;
bacaWarnaRaw(tempR, tempG, tempB);

if (tempR > 0 && tempG > 0 && tempB > 0) {
    sumR += tempR;
    sumG += tempG;
    sumB += tempB;
    validSamples++;
}
delay(10);
}

if (validSamples > 0) {
    r = sumR / validSamples;
    g = sumG / validSamples;
    b = sumB / validSamples;
} else {
    r = g = b = 0;
}
}

// ===== Klasifikasi Tomat =====

bool isMerah(int r, int g, int b) {
    if (r == 0 && g == 0 && b == 0) {
        Serial.println(">>> Pembacaan tidak valid.");
        return false;
    }

    float total = r + g + b;
    float rRatio = (float)r / total;
    float gRatio = (float)g / total;
    float bRatio = (float)b / total;

```

```

float totalCalRed = thresholds.avgRedR + thresholds.avgRedG +
thresholds.avgRedB;

float totalCalGreen = thresholds.avgGreenR + thresholds.avgGreenG +
thresholds.avgGreenB;

float calRedRRatio = (float)thresholds.avgRedR / totalCalRed;
float calRedGRatio = (float)thresholds.avgRedG / totalCalRed;
float calRedBRatio = (float)thresholds.avgRedB / totalCalRed;

float calGreenRRatio = (float)thresholds.avgGreenR / totalCalGreen;
float calGreenGRatio = (float)thresholds.avgGreenG / totalCalGreen;
float calGreenBRatio = (float)thresholds.avgGreenB / totalCalGreen;

float distToRed = sqrt(pow(rRatio - calRedRRatio, 2) + pow(gRatio -
calRedGRatio, 2) + pow(bRatio - calRedBRatio, 2));

float distToGreen = sqrt(pow(rRatio - calGreenRRatio, 2) + pow(gRatio -
calGreenGRatio, 2) + pow(bRatio - calGreenBRatio, 2));

bool result = distToRed < distToGreen;

Serial.print("RGB: ");
Serial.print(r);
Serial.print(",");
Serial.print(g);
Serial.print(",");
Serial.print(b);
Serial.print(" | Rasio R:");
Serial.print(rRatio, 3);
Serial.print(" G:");
Serial.print(gRatio, 3);
Serial.print(" B:");
Serial.print(bRatio, 3);
Serial.print(" | Jarak Merah: ");
Serial.print(distToRed, 4);

```

```

    Serial.print(" | Jarak Hijau: ");
    Serial.print(distToGreen, 4);
    Serial.print(" | Hasil: ");
    Serial.println(result ? "MERAH" : "HIJAU");

    return result;
}

// ===== State Machine =====
enum SystemState {
    WAITING_FRUIT,
    CONVEYOR_STOPPED,
    FRUIT_CLASSIFIED,
    WAITING_IR2
};

SystemState currentState = WAITING_FRUIT;
unsigned long stateStartTime = 0;

// ===== Loop utama =====
void loop() {
    unsigned long currentTime = millis();
    bool detectedIR1 = (digitalRead(IR_SENSOR1) == LOW);
    bool detectedIR2 = (digitalRead(IR_SENSOR2) == LOW);

    switch (currentState) {
    case WAITING_FRUIT:
        if (detectedIR1) {
            Serial.println(">>> BUAH TERDETEKSI DI IR1");
            digitalWrite(RELAY_CONVEYOR, LOW);
            currentState = CONVEYOR_STOPPED;
            stateStartTime = currentTime;
        }
        break;

```

```
case CONVEYOR_STOPPED:
```

```
    if (currentTime - stateStartTime >= conveyorStopTime) {  
        Serial.println(">>> Memulai pembacaan warna...");  
        int r, g, b;  
        bacaWarna(r, g, b);  
        buahSaatIniMerah = isMerah(r, g, b);  
  
        if (buahSaatIniMerah) {  
            countMasak++;  
            menungguBuahDiIR2 = false;  
            Serial.println(">>> BUAH MERAH - Melewati conveyor");  
        } else {  
            countMentah++;  
            menungguBuahDiIR2 = true;  
            Serial.println(">>> BUAH HIJAU - Akan dibuang di IR2");  
        }  
    }
```

```
    lcd.setCursor(7, 0);  
    lcd.print("  ");  
    lcd.setCursor(7, 0);  
    lcd.print(countMasak);  
    lcd.setCursor(7, 1);  
    lcd.print("  ");  
    lcd.setCursor(7, 1);  
    lcd.print(countMentah);
```

```
    digitalWrite(RELAY_CONVEYOR, HIGH);  
    Serial.println(">>> CONVEYOR MENYALA KEMBALI");  
    currentState = FRUIT_CLASSIFIED;  
}  
break;
```

```
case FRUIT_CLASSIFIED:
```

```

    if (!detectedIR1) {
        currentState = WAITING_IR2;
        Serial.println(">>> MENUJU STATE WAITING_IR2");
    }
    break;

case WAITING_IR2:
    if (detectedIR2 && !prevIR2State) {
        Serial.println(">>> BUAH TERDETEKSI DI IR2!");
        if (menungguBuahDiIR2 && !buahSaatIniMerah && !servoBusy) {
            servoSortir.write(POS_SORTIR_BUANG);
            servoBusy = true;
            servoStartTime = currentTime;
            Serial.println(">>> SERVO MEMBUANG BUAH HIJAU!");
        } else if (buahSaatIniMerah) {
            Serial.println(">>> Buah merah di IR2 - servo tidak bergerak");
        }
    }

    if (!detectedIR2 && prevIR2State) {
        Serial.println(">>> IR2 clear - Menunggu buah baru");
        currentState = WAITING_FRUIT;
        menungguBuahDiIR2 = false;
    }
    break;
}

prevIR2State = detectedIR2;

if (servoBusy && currentTime - servoStartTime >= servoDurasi) {
    servoSortir.write(POS_SORTIR_AWAL);
    servoBusy = false;
    Serial.println(">>> SERVO KEMBALI KE POSISI AWAL");
}

```

```

static unsigned long lastDebugTime = 0;
if (currentTime - lastDebugTime >= 3000) {
    lastDebugTime = currentTime;
    Serial.print("State: ");
    switch (currentState) {
    case WAITING_FRUIT:
        Serial.println("WAITING_FRUIT");
        break;
    case CONVEYOR_STOPPED:
        Serial.println("CONVEYOR_STOPPED");
        break;
    case FRUIT_CLASSIFIED:
        Serial.println("FRUIT_CLASSIFIED");
        break;
    case WAITING_IR2:
        Serial.println("WAITING_IR2");
        break;
    }
    Serial.print("Buah Merah: ");
    Serial.print(buahSaatIniMerah);
    Serial.print(" | Menunggu di IR2: ");
    Serial.print(menungguBuahDiIR2);
    Serial.print(" | Servo Busy: ");
    Serial.println(servoBusy);
    Serial.print("IR1: ");
    Serial.print(detectedIR1 ? "DETECTED" : "CLEAR");
    Serial.print(" | IR2: ");
    Serial.println(detectedIR2 ? "DETECTED" : "CLEAR");
    Serial.println("-----");
}
}

```


2.5 Teknik Analisis Data

Pada penelitian ini, data hasil pengujian prototype penyortiran buah tomat berdasarkan warna dianalisis menggunakan Teknik analisis data kuantitatif deskriptif. Analisis ini dilakukan dengan cara menghitung jumlah keberhasilan dan kegagalan sensor dalam mengklasifikasikan buah tomat dengan Tingkat kematangannya.

Data diuji yang diperoleh kemudian disajikan dalam bentuk table distribusi frekuensi dan presentase, sehingga dapat menggambarkan sejauh mana prototype mampu mendeteksi warna tomat dengan benar. Selain itu, analisis deskriptif juga ditunjukkan melalui perhitungan nilai akurasi, error rate, serta penyajian dalam bentuk table untuk memperjelas perbandingan antara jumlah keberhasilan dan kegagalan. Hasil analisis ini memberikan Gambaran sejauh mana system prototype yang dirancang sudah bekerja sesuai dengan tujuan, serta bagian mana yang masih memerlukan perbaikan.

Dengan demikian, Teknik analisis data kuantitatif deskriptif pada penelitian ini berfungsi untuk mendeskripsikan performa sensor TCS3200 dalam penyortiran buah tomat berdasarkan warna secara sederhana, terukur dan mudah dipahami.

BAB III

HASIL DAN PEMBAHASAN

3.1 Hasil pengujian Alat

Pengujian dilakukan dengan menggunakan sejumlah buah tomat yang memiliki Tingkat kematangan berbeda : mentah dan matang. Setiap tomat diletakkan satu per satu pada jalur konveyor untuk diuji oleh sistem sortir otomatis.



Proses kerja sistem dimulai saat sensor infrared mendeteksi keberadaan tomat. Sensor TCS3200 kemudian membaca nilai warna (RGB) dari permukaan tomat. Data yang diperoleh diolah oleh Arduino Uno R3 untuk menentukan tingkat kematangan. Berdasarkan hasil klasifikasi warna, motor servo diarahkan untuk mengarahkan tomat ke jalur yang sesuai.



3.2 Pembahasan

Pada pengujian akurasi data menggunakan metode confusion matri untuk mengetahui apakah data atau nilai yang diperoleh dari prototipe penyortiran buah tomat sudah akurat dan masuk kedalam klasifikasi data yang ditentukan dengan menggunakan data uji sebanyak 4 objek buah tomat.

Setelah dilakukan pengujian akurasi data terhadap prototipe pada penelitian ini, maka didapatkan akurasi data seperti berikut :

- Tabel pengujian akurasi data

No	Objek Buah Tomat	Klasifikasi pada prototype	Uji ke-1	Uji ke-2	Uji ke-3	Uji ke-4	Uji ke-5	Keterangan
1	 Matang	Merah	Sesuai	Sesuai	Gagal	Sesuai	Gagal	Gagal pada uji ke-3 dan ke-5 karena pencahayaan berubah
2	 Matang	Merah	Gagal	Sesuai	Sesuai	Sesuai	Sesuai	Gagal pada pengujian pertaman

								karena posisi tomat miring
3	 Mentah	Hijau	Gagal	Sesuai	Sesuai	Sesuai	Sesuai	Gagal pada pengujian pertama karena warna mendekati batas threshold (hijau kekuningan)
4	 Mentah	Hijau	Sesuai	Sesuai	Sesuai	Sesuai	Sesuai	Stabil

Tabel 1 : pengujian data

Keterangan :

1. Total data testing = 4 Buah
2. Data matang = 2 Buah
3. Data mentah = 2 Buah

- Table perhitungan confusion matrix

Confusion matrix	Matang	Mentah
Matang	7 (True Positive)	3 (False Negative)
Mentah	1 (False Positive)	9 (True Negative)

Tabel 2 : confusion matrix

$$\begin{aligned}
 \text{Error rate} &= \frac{FP+FN}{TP+TN+FP+FN} \times 100\% \\
 &= \frac{1+3}{20} \times 100\% = 20\%
 \end{aligned}$$

$$\begin{aligned}
 \text{Akurasi} &= \frac{TP+TN}{TP+TN+FP+FN} \times 100\% \\
 &= \frac{7+9}{20} \times 100\% \\
 &= 80\%
 \end{aligned}$$

Jadi Tingkat akurasi yang didapatkan dari prototipe penyortiran buah tomat adalah sebesar 80%.

3.3 Pengujian blackbox

Pengujian blackbox merupakan salah satu cara untuk melakukan pengujian terhadap prototipe penyortiran buah tomat berdasarkan Tingkat kematangan menggunakan sensor warna TCS3200 untuk menghasilkan output yang sesuai dengan yang diinginkan. Adapun cara kerja prototipe pada pengujian ini Adalah sebagai berikut :

3.3.1 Pengujian komponen perangkat

Pada tahap ini, pengujian akan dilakukan pada komponen yang terdapat pada protipe penyortiran buah tomat, dapat dilihat pada gambar berikut :



gambar 11: alat prototype



Pada gambar dapat dijelaskan bahwa prototipe penyortiran buah tomat telah selesai dalam tahap perakitan, tiap-tiap komponen sudah berfungsi dan sudah dapat digunakan. Adapun penjelasan hasil pengujian terhadap komponen prototipe penyortiran buah tomat dapat dilihat pada table berikut :


No	Komponen yang di uji	Skenario pengujian	Hasil yang diharapkan	Hasil
1	Sensor warna TCS3200	Berfungsi menyensor objek	Dapat menyendor warna objek dan mengirim nilai	[✓] Sesuai [] Tidak sesuai

			hasil sensor ke mikrokontroller.	
2	Motor DC	Berfungsi menggerakkan conveyor	Dapat menggerakkan conveyor.	[✓] Sesuai [] Tidak sesuai
3	Motor servo	Berfungsi menggerakkan tuas penyortir	Dapat memisahkan/menyortir buah yang matang/mentah	[✓] Sesuai [] Tidak sesuai
4	Arduino Uno R3	Berfungsi memberi perintah	Dapat memproses perintah/dapat menjadi mikrokontroller	[✓] Sesuai [] Tidak sesuai
5	Sensor IR	Berfungsi mendeteksi objek	Berhasil mendeteksi adanya objek pada konveyor	[✓] Sesuai [] Tidak sesuai

Tabel 3 : pengujian perangkat

3.3.2 Pengujian data hasil sensor

Pada tahap ini, setelah prototipe berhasil dijalankan maka dapat memisahkan antara buah yang matang atau mentah. Contoh peletakan buah tomat dapat dilihat pada gambar berikut :



gambar 12 : pengujian alat

Pada gambar dapat dilihat buah tomat berada di konveyor lalu akan dibawa untuk disensor dan nilai hasil sensor akan dikirim ke mikorkontroller, setelah itu konveyor akan membawa buah yang sudah disensor ke alat penyortir (servo) yang akan mengarahkan buah ke kotak klasifikasi mentah/matang.

Ketika buah yang sudah disensor dibawa oleh konveyor ke alat sortir yang akan mengarahkan objek ke kotak kotak klasifikasi buah yang matang/mentah sesuai dengan hasil nilai sensor yang didapat.

Adapun penjelasan hasil pengujian prototipe penyortiran buah tomat dapat dilihat pada tabel :

No	Komponen yang di uji	Skenario pengujian	Hasil pengujian yang diharapkan	Hasil
1	Buah tomat yang sudah berwarna merah	Buah berjalan menuju sensor lalu ke kotak kalsifikasi matang	Buah masuk ke kotak klasifikasi matang	[✓] Sesuai [] Tidak sesuai
2	Buah tomat yang belum berwarna merah	Buah berjalan menuju sensor lalu ke kotak kalsifikasi mentah	Buah masuk ke kotak klasifikasi mentah	[✓] Sesuai [] Tidak sesuai

Tabel 4 : hasil akhir pengujian

BAB IV

KESIMPULAN DAN SARAN

4.1 Kesimpulan

Dari hasil penelitian yang telah dilakukan, dapat disimpulkan bahwa prototipe penyortiran tomat otomatis berbasis sensor warna TCS3200 berhasil dirancang dan diimplementasikan dengan baik. Sistem ini mampu mendeteksi keberadaan tomat melalui sensor infrared dan mengklasifikasikan tingkat kematangan berdasarkan nilai warna RGB yang dibaca oleh sensor warna. Arduino Uno R3 berperan sebagai pusat pengendali yang mengatur proses sortir, menggerakkan motor servo, serta menampilkan informasi hasil klasifikasi pada LCD. Hasil pengujian menunjukkan bahwa alat ini dapat bekerja secara otomatis dan memberikan hasil sortir yang lebih cepat dan konsisten dibandingkan metode manual. Penggunaan komponen sederhana namun fungsional menjadikan prototipe ini cukup efektif untuk diterapkan dalam skala kecil sebagai bentuk penerapan teknologi tepat guna di bidang pertanian.

4.2 Saran

Untuk pengembangan lebih lanjut, sistem ini masih memiliki potensi untuk disempurnakan. Perlu dilakukan pengujian dengan jenis tomat yang lebih bervariasi agar sistem dapat menangani kondisi nyata di lapangan secara lebih akurat. Selain itu, pengaruh pencahayaan terhadap hasil pembacaan sensor juga perlu diperhatikan agar sistem tetap stabil meskipun digunakan di lingkungan berbeda. Penambahan fitur sortir berdasarkan ukuran atau berat dapat menjadi nilai tambah untuk meningkatkan fungsi alat. Diharapkan penelitian ini menjadi dasar bagi pengembangan alat sortir otomatis lainnya yang dapat membantu petani dan pelaku usaha pertanian dalam meningkatkan kualitas dan efisiensi kerja.

Daftar Pustaka

- [1] M. F. Amin, S. R. Akbar, dan E. R. Widasari, “Rancang Bangun Sistem Sortir Buah Apel Menggunakan Sensor Warna dan Sensor Suhu,” *Jurnal Internetworking Indonesia*, vol. 1, no. 3, pp. 45–50, 2017.
- [2] E. Muchyar, Asniati, dan Wiwin, “Sistem Kontrol Otomatis Pada Penyortiran Buah Tomat Menggunakan Mikrokontroler Atmega 2560,” *Jurnal Teknologi dan Sistem Komputer*, vol. 5, no. 2, pp. 134–140, 2017.
- [3] A. Supriyanto dan F. Nugroho, “Penggunaan Motor DC pada Sistem Otomatisasi Sortir Buah Menggunakan Sensor Warna dan Arduino Uno,” *Jurnal Teknologi Informasi dan Komputer*, vol. 6, no. 2, pp. 87–93, 2020.
- [4] R. H. Kusuma dan R. Siregar, “Penggunaan Modul Relay Pada Sistem Otomatisasi Berbasis Arduino,” *Jurnal Sistem Elektronika*, vol. 3, no. 2, pp. 98–104, 2018.